

NAG Library Routine Document

G05MKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05MKF generates a vector of pseudorandom integers from the discrete Poisson distribution with mean λ .

2 Specification

```
SUBROUTINE G05MKF(MODE, LAMBDA, N, X, IGEN, ISEED, R, NR, IFAIL)
INTEGER          MODE, N, X(N), IGEN, ISEED(4), NR, IFAIL
double precision LAMBDA, R(NR)
```

3 Description

G05MKF generates n integers x_i from a discrete Poisson distribution with mean λ , where the probability of $x_i = I$ is

$$P(x_i = I) = \frac{\lambda^I \times e^{-\lambda}}{I!}, \quad I = 0, 1, \dots,$$

where $0 \leq \lambda$.

The variates can be generated with or without using a search table and index. If a search table is used then it is stored with the index in a reference vector and subsequent calls to G05MKF with the same parameter values can then use this reference vector to generate further variates. The reference array is found using a recurrence relation if λ is less than 50 and by Stirling's formula otherwise.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05MKF.

4 References

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

1: MODE – INTEGER *Input*

On entry: a code for selecting the operation to be performed by the routine.

MODE = 0

Set up reference vector only.

MODE = 1

Generate variates using reference vector set up in a prior call to G05MKF.

MODE = 2

Set up reference vector and generate variates.

MODE = 3

Generate variates without using the reference vector.

Constraint: MODE = 0, 1, 2 or 3.

- 2: LAMBDA – *double precision* *Input*
On entry: λ , the mean of the Poisson distribution.
Constraint: LAMBDA ≥ 0.0 .
- 3: N – INTEGER *Input*
On entry: n , the number of pseudorandom numbers to be generated.
Constraint: N ≥ 1 .
- 4: X(N) – INTEGER array *Output*
On exit: the n pseudorandom numbers from the specified Poisson distribution.
- 5: IGEN – INTEGER *Input*
On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.
- 6: ISEED(4) – INTEGER array *Communication Array*
On entry: contains values which define the current state of the selected generator.
On exit: contains updated values defining the new state of the selected generator.
- 7: R(NR) – *double precision* array *Communication Array*
On entry: if MODE = 1, the reference vector from the previous call to G05MKF.
On exit: the reference vector.
- 8: NR – INTEGER *Input*
On entry: the dimension of the array R as declared in the (sub)program from which G05MKF is called.
Suggested value: NR = $22 + 20 \times \sqrt{\text{LAMBDA}}$.
Constraints:
 if MODE = 0 or 2,
 NR > $\left(\text{int} \left[\text{LAMBDA} + 7.15\sqrt{\text{LAMBDA}} + 8.5 \right] - \max \left(0, \text{int} \left[\text{LAMBDA} - 7.15\sqrt{\text{LAMBDA}} \right] \right) + 6 \right)$;
 if MODE = 1, NR should remain unchanged from the previous call to G05MKF;
 if MODE = 3, R is not referenced.
- 9: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by $X04AAF$).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $N < 1$.

$IFAIL = 2$

On entry, NR is too small when $MODE = 0$ or 2 (see Section 5).

$IFAIL = 3$

On entry, $LAMBDA < 0.0$.

$IFAIL = 4$

On entry, $MODE \neq 0, 1, 2$ or 3 .

$IFAIL = 5$

$MODE = 0$ or 2 and $LAMBDA$ is such that NR would have to be larger than the largest representable integer. Use $MODE = 3$ in this case.

$IFAIL = 6$

$LAMBDA$ is not the same as when R was set up in a previous call with $MODE = 0$ or 2 .

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints 10 pseudorandom integers from a Poisson distribution with mean $\lambda = 20$, generated by a single call to $G05MKF$, after initialization by $G05KBF$.

9.1 Program Text

```
*      G05MKF Example Program Text
*      Mark 20 Release. NAG Copyright 2001.
*      .. Parameters ..
      INTEGER          NOUT, N, NR
      PARAMETER        (NOUT=6,N=10,NR=120)
*      .. Local Scalars ..
      DOUBLE PRECISION LAMBDA
      INTEGER          I, IFAIL, IGEN, MODE
*      .. Local Arrays ..
      DOUBLE PRECISION R(NR)
      INTEGER          ISEED(4), X(N)
*      .. External Subroutines ..
      EXTERNAL        G05KBF, G05MKF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05MKF Example Program Results'
      WRITE (NOUT,*)
*      Set the distribution parameter LAMBDA
      LAMBDA = 20.0D0
```

```

*      Initialize the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 42344
      ISEED(4) = 742355
*      IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
      MODE = 0
      IFAIL = 1
*      Generate reference vector R
      CALL G05MKF(MODE,LAMBDA,N,X,IGEN,ISEED,R,NR,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
        MODE = 1
*      Generate integers and store in X
        IFAIL = 0
        CALL G05MKF(MODE,LAMBDA,N,X,IGEN,ISEED,R,NR,IFAIL)
*
        WRITE (NOUT,99999) (X(I),I=1,N)
      ELSE
        WRITE (NOUT,99998) IFAIL
      END IF
*
99999 FORMAT (1X,I12)
99998 FORMAT (1X,' ** G05MKF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

None.

9.3 Program Results

G05MKF Example Program Results

```

14
28
19
23
28
19
21
19
18
26

```
